Numerical Integration Methods.

Kenneth Levasseur
Mathematical Sciences
UMass Lowell
Kenneth_Levasseur@uml.edu

Preliminaries

See http://demonstrations.wolfram.com/ComparingBasicNumericalIntegrationMethods/

```
a = 0;
b = 1;
f[x_] := 1 / (x + 1) ^ 2;
c = (b + a) / 2;
tline = (f[b] - f[a]) / (b - a) (x - a) + f[a];
fp = Plot[f[x], {x, a, b}, PlotRange → {0, 1.2}, AxesOrigin → {0, 0}, Filling → None];
tp = Plot[tline, {x, a, b}, PlotRange → {0, 1.2}, AxesOrigin → {0, 0},
  Filling → Axis, FillingStyle → Opacity[0.3]]; Show[{fp, tp}]
```



What does this image say about the relationship between the "trapazoid error" and concavity?

In[18]:=
```
Manipulate[a = 0;
 b = 1;
 f[x_] := 1 / (x + 1) ^ 2;
 c = (b + a) / 2;
 line = m (x - c) + f[c];
 fp = Plot[f[x], {x, a, b}, PlotRange → {0, 1.2}, AxesOrigin → {0, 0}, Filling → None];
 lp = Plot[line, {x, a, b}, PlotRange → {0, 1.2},
   AxesOrigin → {0, 0}, Filling → Axis, FillingStyle → Opacity[0.3]];
 Show[{fp, lp}], {{m, 0}, -3 f'[c], 3 f'[c]}]
```

Out[18]=



What does this animation say about the relationship between the “midpoint error” and concavity?

Analysis of Integration Errors

- **Definition of IntegerationError function**

```
IntegrationError[f_, a_, b_] := Module[{true, m, d, mid, trap},
  true = Integrate[f, {x, a, b}] // N; m = (a + b) / 2; d = b - a;
  mid = d (f /. {x → m}) // N; trap = d ((f /. {x → a}) + (f /. {x → b})) / 2 // N;
  {f, a, b, true, mid, trap, true - mid, true - trap}]
```

```
heads = {"f(x)", "a", "b", "exact", "midpoint", "trapazoid", "mid error", "trap.error"};
```

■ Pool of test functions

```
pool = {{1 / x, 1, 2.}, {E^x, -1, 1.}, {Cos[x], 0.0, Pi / 2}, {x^3, 0, 1},
  {x^4, 0, 1}, {Log[x], 1, 5}, {2^x, 0, 2}, {ArcSin[x], 0, 1 / 2}, {Sqrt[x], 1, 4}}
```

$\frac{1}{x}$	1	2.
e^x	−1	1.
$\cos(x)$	0.	$\frac{\pi}{2}$
x^3	0	1
x^4	0	1
$\log(x)$	1	5
2^x	0	2
$\sin^{-1}(x)$	0	$\frac{1}{2}$
$\sqrt{x}$	1	4

■ Results

```
(results = (IntegrationError @@ #) & /@ pool) // Prepend[#, heads] &
```

$f(x)$	a	b	exact	midpoint	trapazoid	mid error	trap.error
$\frac{1}{x}$	1	2.	0.693147	0.666667	0.75	0.0264805	−0.0568528
e^x	−1	1.	2.3504	2.	3.08616	0.350402	−0.735759
$\cos(x)$	0.	$\frac{\pi}{2}$	1.	1.11072	0.785398	−0.110721	0.214602
x^3	0	1	0.25	0.125	0.5	0.125	−0.25
x^4	0	1	0.2	0.0625	0.5	0.1375	−0.3
$\log(x)$	1	5	4.04719	4.39445	3.21888	−0.34726	0.828314
2^x	0	2	4.32809	4.	5.	0.328085	−0.671915
$\sin^{-1}(x)$	0	$\frac{1}{2}$	0.127825	0.12634	0.1309	0.00148466	−0.0030749
$\sqrt{x}$	1	4	4.66667	4.74342	4.5	−0.0767498	0.166667

■ Extract Error Data

```
data = results // Transpose // Take[#, -2] & // Transpose // N
```

0.0264805	−0.0568528
0.350402	−0.735759
−0.110721	0.214602
0.125	−0.25
0.1375	−0.3
−0.34726	0.828314
0.328085	−0.671915
0.00148466	−0.0030749
−0.0767498	0.166667

```
ListPlot[data, PlotStyle → Red]
```



■ Fit data to a line

```
bf = Fit[data, {x, 1}, x]
```

0.0155952 − 2.18408 x

```
Plot[bf, {x, -0.5, 0.5}, Epilog → {PointSize[0.02], RGBColor[1, 0, 0], Point /@ data}]
```



■ Ratios of errors

```
Clear[a, b];
data /. {{a_, b_} → {1, b / a}}
```

$$\begin{pmatrix} 1 & -2.14697 \\ 1 & -2.09975 \\ 1 & -1.93823 \\ 1 & -2. \\ 1 & -2.18182 \\ 1 & -2.38529 \\ 1 & -2.04799 \\ 1 & -2.07111 \\ 1 & -2.17156 \end{pmatrix}$$

Let's approximate this ratio with 2.   What does this observation about the ratios of errors say in terms of the following quantities?

I = The exact integral

T = The trapazoid approximation of the integral

M = The midpoint approximation of the integral

Use the relationship you get to find an approximate value of I.

Integration by fitting a quadratic to a function at the endpoints and the midpoint of an interval.

Fit a quadratic to the points $\{a, f(a)\}, \left\{\frac{a+b}{2}, f\left(\frac{a+b}{2}\right)\right\}, \{b, f(b)\}$. There is lots of messy algebra, that is somewhat simpler to do by hand, but here is the result.

```
Clear[f, a, b]
p = InterpolatingPolynomial[{ {a, f(a)}, {(a + b)/2, f((a + b)/2)}, {b, f(b)}}, x] // Simplify
```

$$\frac{1}{(a-b)^2}\left(f(a)(b-x)(a+b-2x)+(a-x)\left(f(b)(a+b-2x)+4(x-b)f\left(\frac{a+b}{2}\right)\right)\right)$$

What is remarkable is that if you integrate this quadratic you get Simpson's Rule. Notice that $-\frac{1}{6}(a-b) = \frac{b-a}{6}$

```
Integrate[p, {x, a, b}]
```

$$-\frac{1}{6}(a-b)\left(4f\left(\frac{a+b}{2}\right)+f(a)+f(b)\right)$$

■ Simpson’s Rule, as releted to the trapazoid and midpoint rules

For any positive integer n,

$$S_{2n} = \frac{1}{3} T_n + \frac{2}{3} M_n.$$

Error Estimates for Trapazoid, Midpoint and Simpson’s rules

The following error estimates appear in virtually every calculus textbook in more or less the same form. The main variation is that what we call S_{2n} is occasionally called S_n, which makes the estimate for E_S look different.

Suppose $|f''(x)| \leq K$ for $a \leq x \leq b$ then the errors in approximation of $\int_a^b f(x)\,dx$ with n uniform subintervals using the Trapazoid and Midpoints rules satisfy the bounds

$$|E_T| \leq \frac{K(b-a)^3}{12n^2} \quad \text{and} \quad |E_M| \leq \frac{K(b-a)^3}{24n^2}$$

If $\left|f^{(4)}(x)\right| \leq K'$ then $|E_S| \leq \frac{K'(b-a)^5}{180n^4}$ for even n.

“Hermite Integration?”

Consider approximating a function on each subinterval by the polynomial that agrees with the function and it’s first derivative on the endpoints and the midpoint.

6 conditions $\Rightarrow$ the polynomial will have degree 5

Mathematica has a function that does this.

In[1]:=
```
? InterpolatingPolynomial
```

InterpolatingPolynomial[{f_1, f_2, …}, x] constructs an interpolating polynomial in x which reproduces the function values f_i at successive integer values 1, 2, … of x.
InterpolatingPolynomial[{{x_1, f_1}, {x_2, f_2}, …}, x] constructs an interpolating polynomial for the function values f_i corresponding to x values x_i.
InterpolatingPolynomial[{{{x_1, y_1, …}, f_1}, {{x_2, y_2, …}, f_2}, …}, {x, y, …}] constructs a multidimensional interpolating polynomial in the variables x, y, ….
InterpolatingPolynomial[{{{x_1, …}, f_1, df_1, …}, …}, {x, …}] constructs an interpolating polynomial that reproduces derivatives as well as function values. ≫

In[2]:=
```
y =
 InterpolatingPolynomial[{{{0}, 0, 1}, {{Pi / 4}, √2 / 2, √2 / 2}, {{Pi / 2}, 1, 0}}, {x}] // N
```

Out[2]= $x((((0.00572168(x-1.5708)+0.0237772)(x-0.785398)-0.151618)(x-0.785398)-0.126921)x+1.)$

In[3]:=
```
Plot[{y, Sin[x]}, {x, 0, Pi / 2}]
```

Out[3]=



In[4]:=
```
Plot[{y - Sin[x]}, {x, 0, Pi / 2}]
```

Out[4]=



In[5]:=
```
HermiteIntegrate[f_, {a_, b_}, x_, 1] := Module[{y},
  y = InterpolatingPolynomial[{{{a}, f[a], f'[a]}, {{(a + b)/2}, f[(a + b)/2], f'[(a + b)/2]},
     {{b}, f[b], f'[b]}}, {x}]; Integrate[y, {x, a, b}]]
```

In[6]:=
```
Clear[f, a, b]
```

In[7]:=
```
HermiteIntegrate[f, {a, b}, x, 1] // Simplify
```

Out[7]= $\frac{1}{60}(a-b)\left((a-b)(f'(a)-f'(b))-32 f\left(\frac{a+b}{2}\right)-14 f(a)-14 f(b)\right)$

In[8]:=
```
HermiteIntegrate[Sin, {0, Pi / 2}, x, 1] // N
```

Out[8]= 1.00003

In[9]:=
```
g = Sin[#^2] &
```

Out[9]= $\sin\left(\#1^2\right)$ &

In[10]:=
```
{HermiteIntegrate[g, {0., 1.}, x, 1] // N, NIntegrate[g[x], {x, 0., 1.}]}
```

Out[10]= {0.310282, 0.310268}

In[11]:=
```
HermiteIntegrate[f_, {a_, b_}, x_, n_] := Module[{d}, d = (b - a)/n; Total[
    HermiteIntegrate[f, {a + (# - 1) d, a + # d}, x, 1] & /@ Range[n]]] /; (IntegerQ[n] && n > 1)
```

In[12]:=
```
HermiteIntegrate[Sin, {0., Pi / 2}, x, 5] // N
```

Out[12]= 1.

In[13]:=
```
HermiteIntegrate[g, {0., 1.}, x, 5] - NIntegrate[g[x], {x, 0., 1.}]
```

Out[13]= -8.9376×10^{-9}

In[14]:=
```
HermiteIntegrate[f, {a, b}, x, 4] // Simplify
```

Out[14]= $\frac{1}{960}(a-b)\left(-b f'(a)-a f'(b)-128 f\left(\frac{5a}{8}+\frac{3b}{8}\right)-128 f\left(\frac{3a}{8}+\frac{5b}{8}\right)-112 f\left(\frac{a+b}{2}\right)-112 f\left(\frac{1}{4}(3a+b)\right)-128 f\left(\frac{1}{8}(7a+b)\right)-112 f\left(\frac{1}{4}(a+3b)\right)-128 f\left(\frac{1}{8}(a+7b)\right)+a f'(a)-56 f(a)+b f'(b)-56 f(b)\right)$

In[15]:=
```
HermiteIntegrate[f, {0, 1}, x, 4] // Simplify
```

Out[15]= $\frac{1}{960}\left(f'(0)-f'(1)+56 f(0)+128 f\left(\frac{1}{8}\right)+112 f\left(\frac{1}{4}\right)+128 f\left(\frac{3}{8}\right)+112 f\left(\frac{1}{2}\right)+128 f\left(\frac{5}{8}\right)+112 f\left(\frac{3}{4}\right)+128 f\left(\frac{7}{8}\right)+56 f(1)\right)$